PHARMALEX
CONFIDENCE BEYOND COMPLIANCE

Pharmacokinetic Model with SAS
Proc MCMC with applications to preclinical pharmacology

Maud Hennion | Senior Manager Statistics - Manager, Pharmacometrics

Bruno Boulanger | CSO

BAYES
BAYESIAN BIOSTATISTICS
LYON 2019
21-24 MAY

Pharmacokinetic and Compartmental Models

What’s the effect of a drug ?



Black Box



Effects

Drug:
- Dose
- Administration
- Formulation
- Frequency

Model the concentration and the effect

Drug:
- Dose
- Administration
- Formulation
- Frequency





Pharmaco**kinetic** (PK) Pharmaco**dynamic** (PD)



PK/PD

PK – PD: Definition

- **Pharmacokinetic** describes the drug concentration-time courses in body fluids resulting from administration of a certain drug dose

“ What the body does to the drug? “

- **Pharmacodynamic** describes the observed effect resulting from a certain drug concentration

“ What the drug does to the body? “

- The rationale for **PK-PD modelling** is to link pharmacokinetic and pharmacodynamic in order to establish and evaluate dose-concentration-response relationships and subsequently describe and predict the effect-time courses resulting from a drug dose

Compartmental Model in Pharmacokinetic

- **Compartmental models** use kinetic models to describe (with approximations) the concentration-time curve. Models take into account

  - The type of **administration**
    - Oral Administration
    - IV Bolus
    - Infusion
    - …



(a) Intravenous (iv) bolus dose (b) Oral dose

  - The **mechanism** of absorption, distribution and elimination

  - The **tissues type**: A compartment is a region of the body in which the drug is well mixed and kinetically homogenous.
    - Bloodstream (= central compartment)
    - Poorly-perfused tissues (muscle and skin)
    - Brain



Examples of PK compartmental models

$C_1(t) \sim N(\mu = \frac{A_1(t)}{V_1}, \sigma^2)$ **where**

➔ Two compartments model with iv bolus administration



$$\frac{dA_1(t)}{dt} = -k_{el}A_1(t) - k_{12}A_1(t) + k_{21}A_2(t)$$

$$\frac{dA_2(t)}{dt} = k_{12}A_1(t) - k_{21}A_2(t)$$

$$A_1(0) = dose, A_2(0) = 0$$

➔ One compartment model with oral administration



Between-subjects variability can be estimated on the PK parameters:

$$k_{12} = \exp(\beta_1 + b_1), b_1 \sim N(0, \sigma_1^2)$$

➔ **PopPK Models**

$$\frac{dA_0(t)}{dt} = -k_a A_0(t)$$

$$\frac{dA_1(t)}{dt} = \underbrace{k_a A_0(t)}_{\text{Absorption from the gut}} - \underbrace{k_{el}A_1(t)}_{\text{Elimination}}$$

Different parametrization are allowed in PK models

- Parametrization in terms of **elimination and transfer rate constants**
  - k_{el} is the elimination rate
  - k_{12}, k_{21} ... are the transfer rates
- Parametrization in terms of **clearance and volumes constants**
  - **Clearance** (CL) is the theoretical volume of plasma that's completely cleared of drug per unit of time
  - **Volume of distribution** is the theoretical volume that would be necessary to contain the total amount of an administered drug at the same concentration that it is observed in the blood plasma

Elimination and transfer rate	Clearance and volume
k_{el}	CL_1/V_1
k_{12}	CL_2/V_1
k_{21}	CL_2/V_2

Compartmental Models in Proc MCMC
Example on Preclinical Data

Preclinical data
IV and Oral Administration – Single Dose



How to fit Compartmental Model in Proc MCMC (SAS)?

- `call ode` and **`proc fcmp`**
  - **`proc fcmp`** provides the ability to build functions (ODEs in this case).
    Use CALL routines and subroutines to use them in other SAS procedures or DATA steps.
  - `call ode` subroutine numerically solves a set of first-order ordinary differential equations (ODEs), including piecewise differential equations.
    The subroutine is included in the proc mcmc and calls the function built in proc fcmp.

- The `CMPTMODEL` statement computes predicted concentrations from a specified one-, two-, or three-compartment model.
  - Only available in SAS 9.4 - SAS/STAT 15.1
  - Only compartmental model
  - More efficient than call ode
  - Available with proc MCMC and proc NLMIXED

CMPTMODEL Statement

- The CMPTMODEL statement includes three types of options:
  - ***Required-options***: PCONC=, TIME=, NCOMPS=, ADMTYPE=, and PARMTYPE=
    - PCONC= outcome variable that is the predicted concentration in the first/central compartment
    - ADMTYPE = IVB | INF | ORAL
    - NCOMPS = 1 | 2 | 3
    - PARMTYPE = 1 | 2
      - 1 = parametrizes the compartment model in terms of elimination and transfer rate constants
      - 2 = parametrizes the compartment model in terms of clearance and volumes constants
  - ***Conditionally-required-options*** (depending on the specifications of the *required-options)*: CL*n*=, VOL*n*=, K12=, K21=, K13=, K31=, KA=, K*n*0=, RATE=, and DURN=
    - DURN = duration of the infusion (ADMTYPE = INF)
    - RATE = rate of infusion (ADMTYPE = INF)
    - n = the compartment number
  - ***Optional-options***: DOSE*n*=, SCALE*n*=, PCONC0=, PCONC2=, and PCONC3=.

CMPTMODEL options

- Here is a few examples of the conditionally required options and valid optional options

Model	Required	Conditionally required	Optional
One compartment with bolus dose	NCOMPS = 1 ADMTYPE= IVB PARMTYPE = **1**	K10 =	DOSE1= SCALE1=
One compartment with bolus dose	NCOMPS = 1 ADMTYPE=IVB PARMTYPE = 2	CL1 = VOL1 =	DOSE1= SCALE1=
Two compartments with oral dose	NCOMPS = 2 ADMTYPE= ORAL PARMTYPE = 1	K10 = K12= K21= KA =	DOSE1= DOSE2= SCALE1= SCALE2= K20= PCONC2=

Let's fit a compartmental model on the IV data

- Two compartments model with iv bolus administration

$$\frac{dA_1(t)}{dt} = -k_{el}A_1(t) - k_{12}A_1(t) + k_{21}A_2(t)$$

$$\frac{dA_2(t)}{dt} = k_{12}A_1(t) - k_{21}A_2(t)$$

$$A_1(0) = dose, A_2(0) = 0$$

- **Solution 1** : `call ode` and **`proc fcmp`**

```
proc fcmp outlib=sasuser.funcs.PK;
      subroutine TwoComp_IV(t, y[*], dy[*], kel, k12, k21);
      outargs dy;

      dy[1] = -kel*y[1] - k12 * y[1] + k21 * y[2];
      dy[2] = k12 * y[1] - k21 * y[2];

      endsub;
run;
options cmplib=sasuser.funcs;
```

Proc MCMC with call ode

```
proc mcmc data=out nmc=300000 nbi =55000 thin = 30;
          array init[2] dose 0;
          array sol[2];
          array mub[2] beta1 beta2;
          array b[2] b1 b2;
          array covb[2,2];
          array S[2,2] (1 0 0 1);

          parms beta1 -6 beta2 -7 beta3 -5 beta4 -5;
          parms n 5;
          parms covb ;

          prior beta: ~ normal(-5, var = 5);
          prior n  ~ general(0, lower=0);
          prior covb ~ iWish(2, S);

          random b ~ mvn(mub, covb) subject=id;

          CL1  = exp(b1);
          V1   = exp(b2);
          CL2  = exp(beta3);
          V2   = exp(beta4);
          kel = cl1/v1;
          k12 = cl2/v1;
          k21 = cl2/v2;

          call ode('TwoComp_IV', sol, init, 0, time_n, kel, k12, k21);

          m=sol[1]/V1;
          mu = log(m**2/sqrt(n + m**2));
          s2y = log(1 + (n/m**2));
          model concentration ~ lognormal(mu, var=s2y);
run;
```

Random effects on CL1 and V1
➔ **PopPK Model**

Let's fit a compartmental model on the IV data

- One compartment model with iv bolus administration

$$\frac{dA_1(t)}{dt} = -k_{el}A_1(t) - k_{12}A_1(t) + k_{21}A_2(t)$$

$$\frac{dA_2(t)}{dt} = k_{12}A_1(t) - k_{21}A_2(t)$$

$$A_1(0) = dose, A_2(0) = 0$$

- **Solution 2** : `CMPTMODEL` statement

```
CMPTMODEL ncomps = 2 admtype = ivb time = time pconc = sol
          parmtype = 2 cl1=cl1 vol1=v1 cl2=cl2 vol2=v2;

CMPTMODEL ncomps = 2 admtype = ivb time = time pconc = sol
           parmtype = 1 k10=kel k12=k12 k21=k21;
```

Proc MCMC with CMPTMODEL

```
proc mcmc data=out nmc=300000 nbi =55000 thin = 30;

        array mub[2] beta1 beta2;
        array b[2] b1 b2;
        array covb[2,2];
        array S[2,2] (1 0 0 1);

        parms beta1 -6 beta2 -7 beta3 -5 beta4 -5;
        parms n 5;
        parms covb ;

        prior beta: ~ normal(-5, var = 5);
        prior n  ~ general(0, lower=0);
        prior covb ~ iWish(2, S);

        random b ~ mvn(mub, covb) subject=id;

        CL1  = exp(b1);
        V1   = exp(b2);
        CL2  = exp(beta3);
        V2   = exp(beta4);

        CMPTMODEL ncomps = 2 admtype = ivb time = time pconc = sol parmtype = 2 cl1=cl1 vol1=v1 cl2=cl2 vol2=v2;

        m=sol/V1;
        mu = log(m**2/sqrt(n + m**2));
        s2y = log(1 + (n/m**2));
        model dv ~ lognormal(mu, var=s2y);
run;
```

Random effects on CL1 and V1
➔ **PopPK Model**

Output: the two proc mcmc converge to the same solutions











Output: Good fit of the model



- Median of the predictive distribution
- Interval of Prediction (Q2.5%-Q97.5%)

Comparison solution 1 vs solution 2

- Solution 1: `call ode` and **`proc fcmp`**

Posterior Summaries and Intervals					
Parameter	N	Mean	Standard Deviation	95% HPD Interval	
beta1	10000	-5.8152	0.7539	-7.2824	-4.2787
beta2	10000	-6.7326	0.6491	-8.1018	-5.4835
beta3	10000	-5.5535	0.1732	-5.8498	-5.1732
beta4	10000	-5.1525	0.5222	-5.9550	-3.8483
n	10000	1062.1	2274.0	26.4717	4574.3
covb1	10000	1.0610	2.6701	0.0652	3.2596
covb2	10000	0.1706	2.2066	-1.6935	2.1927
covb3	10000	0.1706	2.2066	-1.6935	2.1927
covb4	10000	1.2592	3.4891	0.0668	4.0322

```
NOTE: Starting optimization.
NOTE: Tuning the proposal distribution.
NOTE: Generating the burn-in samples.
NOTE: Beginning sample generation.
NOTE: Generating diagnostic plots.
NOTE: The above message was for the following BY group:
      Compound=465
NOTE: The data set WORK.OUTPOST has 10000 observations and 21 variables.
NOTE: PROCEDURE MCMC used (Total process time):
      real time           11:37:16.53
      cpu time            10:53:47.73
```

- Solution 2: `CMPTMODEL` statement

Posterior Summaries and Intervals					
Parameter	N	Mean	Standard Deviation	95% HPD Interval	
beta1	10000	-5.7412	0.6577	-7.1463	-4.4559
beta2	10000	-6.7116	0.6330	-8.0233	-5.5203
beta3	10000	-5.5633	0.1616	-5.8406	-5.2104
beta4	10000	-5.1708	0.4512	-5.9571	-4.2065
n	10000	1023.5	2772.5	27.6953	4006.8
covb1	10000	1.0419	2.7118	0.0617	3.2942
covb2	10000	0.1643	1.7318	-1.5313	2.1694
covb3	10000	0.1643	1.7318	-1.5313	2.1694
covb4	10000	1.1794	2.5010	0.0603	3.8721

```
NOTE: Starting optimization.
NOTE: Tuning the proposal distribution.
NOTE: Generating the burn-in samples.
NOTE: Beginning sample generation.
NOTE: Generating predictive samples.
NOTE: Generating diagnostic plots.
NOTE: The data set DATA.OUTPOST_465 has 10000 observations and 20 variables.
NOTE: The data set WORK.PRED has 10000 observations and 21 variables.
NOTE: PROCEDURE MCMC used (Total process time):
      real time           4:16.43
      cpu time            3:22.71
```

Let’s fit a model on the complete dataset !

- One model by administration
  - iv bolus administration: 2 compartments model
  - Oral administration: 1 compartment model

- Parametrizes in terms of clearance and volume constants
- Use the CMPTMODEL statement

```
CMPTMODEL ncomps = 2 admtype = ivb time = time pconc = sol
          parmtype = 2 cl1=cl_iv1 vol1=v_iv1 cl2=cl_iv2 vol2=v_iv2;
```

```
CMPTMODEL ncomps = 1 admtype = oral time = time pconc = sol
         parmtype = 2 cl1 = CL_PO vol1=V_PO ka=ka;
```

SAS code

```
proc mcmc data=out nmc=300000 nbi =55000 thin = 30;
          [ … ]

          F    = exp(beta0);
          CL_IV1   = exp(b1);
          V_IV1    = exp(b2);
          CL_IV2   = exp(beta3);
          V_IV2    = exp(beta4);
          CL_PO   = exp(b1)/F;
          V_PO    = exp(b2)/F;
          ka   = exp(beta5) ;

     if Admin="IV" then do;
               CMPTMODEL ncomps = 2 admtype = ivb time = time pconc = sol
               parmtype = 2 cl1=cl_IV1 vol1=v_IV1 cl2=cl_IV2 vol2=v_IV2;
               m=sol/V_IV1;
     end;

     If Admin="PO" then do;
               CMPTMODEL ncomps = 1 admtype = oral time = time pconc = sol
               parmtype = 2 cl1 = CL_PO vol1=V_PO ka=ka;
               m=sol/V_PO;
     end;

          mu = log(m**2/sqrt(n + m**2));
          s2y = log(1 + (n/m**2));
          model dv ~ lognormal(mu, var=s2y);
run;
```



F is the bioavaibility

Conclusion

- A set of Ordinary Differential Equations (ODEs) can be solved in a **Bayesian** way thanks to
  - `call ode` + **`proc fcmp`**
  - `CMPTMODEL` function
- Compartment PopPK model can be handled by the new function `CMPTMODEL` of **`proc mcmc:`**
  - Ease of implementation: no need to write the system of equations
  - Much faster than `call ode`
  - Multiple administration can be easily modeled *(not shown here)*
    - *%pkconvert* macro should be used to convert the dataset in order to get a structure close to NONMEM structure
  - Limitations:
    - Only available in last version of SAS
    - Only classical PK model are available (1-2-3 compartments)
    - Two possibilities for the parametrization

Acknowledgement

This project has been developed in collaboration with Janssen (Beerse, Belgium).

PHARMALEX
CONFIDENCE BEYOND COMPLIANCE

janssen | PHARMACEUTICAL COMPANIES OF Johnson & Johnson